Theorem 2 (i)

In[17]:=

```
(* Define mu *)
mu = (-8 pf pnf^3 (-1 + ql) ql (-1 + qr) qr + 2 pnf^4 (-1 + ql) ql (-1 + qr) qr -
      2 pnf (-2 + qr + 3 pf qr - pf qr^2 + pf ql^2 (-1 + 4 pf^2 (-1 + qr) qr) +
         ql (1 + pf (3 - 4 qr) - 4 pf^3 (-1 + qr) qr)) + pf (-qr (-2 + pf + pf qr) +
         pf ql^2 (-1 + 2 pf^2 (-1 + qr) qr) - ql (-2 + pf + 4 pf qr + 2 pf^3 (-1 + qr) qr)) + pnf^2
       (-8 + 7 qr - qr^2 + ql^2 (-1 + 12 pf^2 (-1 + qr) qr) + ql (7 + 4 (-1 + 3 pf^2) qr - 12 pf^2 qr^2))) /
    (1 + pf^4 (-1 + ql) ql (-1 + qr) qr - 4 pf pnf^3 (-1 + ql) ql (-1 + qr) qr +
      pnf^4 (-1 + ql) ql (-1 + qr) qr - pf^2 (ql + qr + 2 ql qr) -
      2 pf pnf (ql + qr + 2 (-1 + pf^2) ql qr + 2 pf^2 ql^2 (-1 + qr) qr - 2 pf^2 ql qr^2) +
      pnf^2 (-4 + 3 qr + 6 pf^2 ql^2 (-1 + qr) qr + ql (3 - 2 qr + 6 pf^2 qr - 6 pf^2 qr^2)));
(* Compute mu(Sigma,0) - mu(Sigma/2,Sigma/2) *)
compare =
  Simplify[mu /. {ql → Sigma, qr → 0}] - Simplify[mu /. {ql → Sigma / 2, qr → Sigma / 2}];
(* Specify range of parameters *)
conditions = 0 < Sigma < 1 && 0 < pnf < pf ≤ 1 / 2;
(* Verify if it is possible to have mu(Sigma,0) - mu(Sigma/2,Sigma/2) ≤ 0;
returns false if the difference > 0 for all parameters within the range *)
Reduce[compare ≤ 0 && conditions, {Sigma, pf, pnf}]
```

Out[20]=

```
False
```

Theorem 2 (ii)

In[21]:=

```
(* Take second-order derivative w.r.t. q_r *)
secondDerivative = D[mu, {qr, 2}];
(* Take q_l = 1/2 *)
SOD = Simplify[secondDerivative /. {ql → 1 / 2}];
(* Specify range of parameters *)
conditions = 0 < qr < 1 && 0 < pnf < pf ≤ 1 / 2;
(* Verify if it is possible to have SOD ≥ 0;
returns false if SOD < 0 for all parameters within the range *)
Reduce[SOD ≥ 0 && conditions, {qr, ql, pf, pnf}]
```

Out[24]=

```
False
```