Theorem 1 (i): One-sided Allocation > Balanced Allocation

In[1]:=

```
(* Define mu *)
mu = -pnf^4 ((4 - 3 qr)^2 + ql^2 (9 - 10 qr + 2 qr^2) - 2 ql (12 - 16 qr + 5 qr^2)) -
   pf (qr (-2 + pf + pf qr - 2 pf^2 qr + pf^3 qr) + ql (-2 + pf + 4 pf qr + 6 pf^3 qr^2 - 4 pf^2 qr (1 + qr)) +
      pf ql^2 (1 - 2 pf (1 + 2 qr) + pf^2 (1 + 6 qr + 2 qr^2))) +
   2 pnf^3 ((-4 + 3 qr) (-2 + qr + 2 pf qr) + ql^2 (3 - 2 qr + 2 pf (3 - 6 qr + 2 qr^2)) -
      2 ql (5 - 5 qr + qr^2 + 2 pf (2 - 6 qr + 3 qr^2))) + 2 pnf (2 + (-1 - 3 pf + 2 pf^2) qr +
      (pf + pf^2 - 2 pf^3) qr^2 + ql (-1 + 4 pf^3 (-2 + qr) qr + pf (-3 + 4 qr) + pf^2 (2 + 6 qr - 6 qr^2)) +
      pf ql^2 (1 + pf - 6 pf qr + pf^2 (-2 + 4 qr + 4 qr^2))) - pnf^2 (8 + (-7 - 16 pf + 8 pf^2) qr +
      (1 + 10 pf - 2 pf^2) qr^2 + ql (-7 + 4 qr + pf^2 (8 - 12 qr^2) - 4 pf (4 - 9 qr + 3 qr^2)) +
      ql^2 (1 - 2 pf (-5 + 6 qr) + 2 pf^2 (-1 - 6 qr + 6 qr^2)));
(* Compute mu(Sigma,0) - mu(Sigma/2,Sigma/2) *)
compare =
  Simplify[mu /. {ql → Sigma, qr → 0}] - Simplify[mu /. {ql → Sigma / 2, qr → Sigma / 2}];
(* Specify range of parameters *)
conditions = 0 < Sigma ≤ 1 && 0 < pnf < pf ≤ 1 / 2;
(* Verify if it is possible to have mu(Sigma,0) - mu(Sigma/2,Sigma/2) ≤ 0;
returns false if the difference > 0 for all parameters within the range *)
Reduce[compare ≤ 0 && conditions, {Sigma, pf, pnf}]
```

Out[4]= False

Theorem 1 (ii): Concave in direction (0,1)

In[5]:=

```
(* Take second-order derivative w.r.t. q_r *)
SOD = Simplify[D[mu, {qr, 2}]];
(* Specify range of parameters *)
conditions = 0 < ql < 1 && 0 < qr < 1 && 0 < pnf < pf ≤ 1 / 2;
(* Verify if it is possible to have SOD ≥ 0;
returns false if SOD < 0 for all parameters within the range *)
Reduce[SOD ≥ 0 && conditions, {ql, qr, pf, pnf}]
```

Out[7]= False

Theorem 1 (ii): Concave in direction (1,1)

In[8]:=
```
(* Define mu(q_l+h, q_r+h) *)
muh = Simplify[mu /. {ql → ql + h, qr → qr + h}];
(* Take second-order derivative w.r.t. h *)
SOD = Simplify[D[muh, {h, 2}]];
(* Evaluate SOD at h = 0 *)
SOD = Simplify[SOD /. {h → 0}];
(* Specify range of parameters *)
conditions = 0 < ql < 1 && 0 < qr < 1 && 0 < pnf < pf ≤ 1 / 2;
(* Verify if it is possible to have SOD ≥ 0;
returns false if SOD < 0 for all parameters within the range *)
Reduce[SOD ≥ 0 && conditions, {ql, qr, pf, pnf}]
```

Out[12]=
```
False
```

Theorem 1 (iii): Convex in direction (1,-1)

In[13]:=
```
(* Define mu(q_l, Sigma - q_l) *)
musigma = Simplify[mu /. {qr → Sigma - ql}];
(* Take second-order derivative w.r.t. q_l *)
SOD = Simplify[D[musigma, {ql, 2}]];
(* Specify range of parameters *)
conditions = 0 < ql < 1 && ql < Sigma < ql + 1 && 0 < pnf < pf ≤ 1 / 2;
(* Verify if it is possible to have SOD ≤ 0;
returns false if SOD > 0 for all parameters within the range *)
Simplify[Reduce[SOD ≤ 0 && conditions, {Sigma, ql, pnf, pf}]]
```

Out[16]=
```
False
```