Theorem 5 (i): Concave in direction (0,1)

In[1]:=
```
(* Define mu *)
mu = -p^4 ((4 - 3 br)^2 + bl^2 (9 - 10 br + 2 br^2) - 2 bl (12 - 16 br + 5 br^2)) -
    pf (br (-2 + pf + pf br - 2 pf^2 br + pf^3 br) +
        bl (-2 + pf + 4 pf br + 6 pf^3 br^2 - 4 pf^2 br (1 + br)) +
        pf bl^2 (1 - 2 pf (1 + 2 br) + pf^2 (1 + 6 br + 2 br^2))) +
    2 p^3 ((-4 + 3 br) (-2 + br + 2 pf br) + bl^2 (3 - 2 br + 2 pf (3 - 6 br + 2 br^2)) -
        2 bl (5 - 5 br + br^2 + 2 pf (2 - 6 br + 3 br^2))) +
    2 p (2 + (-1 - 3 pf + 2 pf^2) br + (pf + pf^2 - 2 pf^3) br^2 +
        bl (-1 + 4 pf^3 (-2 + br) br + pf (-3 + 4 br) + pf^2 (2 + 6 br - 6 br^2)) +
        pf bl^2 (1 + pf - 6 pf br + pf^2 (-2 + 4 br + 4 br^2))) -
    p^2 (8 + (-7 - 16 pf + 8 pf^2) br + (1 + 10 pf - 2 pf^2) br^2 +
        bl (-7 + 4 br + pf^2 (8 - 12 br^2) - 4 pf (4 - 9 br + 3 br^2)) +
        bl^2 (1 - 2 pf (-5 + 6 br) + 2 pf^2 (-1 - 6 br + 6 br^2)));
(* Take second-order derivative w.r.t. b_r *)
SOD = Simplify[D[mu, {br, 2}]];
(* Specify range of parameters *)
conditions = 0 < bl < 1 && 0 < br < 1 && 0 ≤ p < pf ≤ 1/2;
(* Verify if it is possible to have SOD ≥ 0;
returns false if SOD < 0 for all parameters within the range *)
Reduce[SOD ≥ 0 && conditions, {bl, br, pf, p}]
```

Out[4]=
```
False
```

Theorem 5 (i): Concave in direction (1,1)

In[5]:=
```
(* Define mu(b_l+h, b_r+h) *)
muh = Simplify[mu /. {bl → bl + h, br → br + h}];
(* Take second-order derivative w.r.t. h *)
SOD = Simplify[D[muh, {h, 2}]];
(* Evaluate SOD at h = 0 *)
SOD = Simplify[SOD /. {h → 0}];
(* Specify range of parameters *)
conditions = 0 < bl < 1 && 0 < br < 1 && 0 ≤ p < pf ≤ 1 / 2;
(* Verify if it is possible to have SOD ≥ 0;
returns false if SOD < 0 for all parameters within the range *)
Reduce[SOD ≥ 0 && conditions, {bl, br, pf, p}]
```

Out[9]=
```
False
```

Theorem 5 (ii): Convex in direction (1,-1)

In[10]:=
```
(* Define mu(b_l, B - b_l) *)
musigma = Simplify[mu /. {br → B - bl}];
(* Take second-order derivative w.r.t. b_l *)
SOD = Simplify[D[musigma, {bl, 2}]];
(* Specify range of parameters *)
conditions = 0 < bl < 1 && bl < B < bl + 1 && 0 ≤ p < pf ≤ 1 / 2;
(* Verify if it is possible to have SOD ≤ 0;
returns false if SOD > 0 for all parameters within the range *)
Simplify[Reduce[SOD ≤ 0 && conditions, {B, bl, p, pf}]]
```

Out[13]=
```
False
```